Lecture Notes:
Functional Programming in Typed Racket

Florian Biermann
fbie@itu.dk

2016-05-26

1 Hello World in Typed Racket

```
;; This is a comment!
;; Tell the run-time, which language to use.
#lang typed/racket

;; Now, print something.
(print "Nihao!")
```

You must always start your code with a `#lang` directive. You can also choose other languages than `typed/racket`, such as `racket`, `plaid` or `slides`. For readability, we will omit all `#lang` directives in the remaining code examples.

2 Expressions

Expressions in Typed Racket are of the form:

```
(f arg1 arg2 ... argn)
```

Operators are also just functions:

`(+ x y)`	$\Rightarrow$	$x + y$
`(> x y)`	$\Rightarrow$	$x > y$
`(/ x y)`	$\Rightarrow$	$\frac{x}{y}$
`(f x y)`	$\Rightarrow$	$f(x, y)$

In Racket this is called an *S-expression*.

3 Local Bindings

Local bindings are very similar to local bindings in for instance Java. However, their scope is very explicit:

```
(let ([x (* 2 16)]
      [y (* 3 17)])
    (print (+ x y)))
```

Note that, after the final parenthesis that closes the `let` expression, you cannot reference `x` or `y`:

```
(let ([x (* 2 16)]
      [y (* 3 17)])
    (print (+ x y)))
(print x)
```

Trying this results in an error:

```
; stdin::5873:
;  Type Checker: missing type for top-level identifier;
;  either undefined or missing a type annotation
;   identifier: x
;   in: x
```

4 Functions and Types

We'd like to define a simple function `times-two` that multiplies its argument `x` by two:

```
(: times-two (-> Number Number))
(define (times-two x)
  (* 2 x))
```

There are two things to note here. The first is that there is no return statement in Racket. Instead, a function evaluates to the result of the last expression in its body. The second is that `:` denotes a *type annotation*. We have annotated `times-two` to be of type $Number \rightarrow Number$.

In a type-expression, the return-type is the last type that occurs. Consider the type `(-> A B C)`. The return-type is `C` and `A` and `B` are parameter types. If the type is `(-> A B A)`, `A` is both parameter and return type. `B` is only parameter type.

We could also have written `times-two`'s type like this:

```
(define (times-two [x : Number])
  (* 2 x))
```

Typed Racket can then infer the correct type of the function. We can ask the REPL[1] for the `times-two`'s type:

```
> times-two
- : (-> Number Number)
#<procedure:times-two>
```

[1] REPL is short for Read-Eval-Print-Loop.

We will often use both notations. The latter notations is in particular used to define the parameter types for `struct` constructors.

5 Conditionals

This is a well-known function:

```
(: abs (-> Number Number)
(define (abs x)
  (if (< x 0)
      (- x)
      x))
```

(Exercise: How else could you have written the type?)

It contains a conditional. Conditionals have the form `(if B ET EF)` where `B` is a boolean expression, `ET` is the expression that is evaluated if `B` evaluates to true (`#t`), `EF` is the expression that is evaluated if `B` evaluates to false (`#f`).

During the lecture, I told you that the types of `ET` and `EF` must be the same. This is true in most functional programming languages. However, Typed Racket is a bit special. Look at this example:

```
(define (funny [b : Boolean])
  (if b "True" 54727565))
```

Let's check the function's type:

```
> funny
- : (-> Boolean (U Positive-Index String))
#<procedure:funny>
```

Instead of raising a type-error, Typed Racket infers from the usage of the conditional, that the return-type of `funny` must be the union of the types of both branches. `U` stands for set-union, $\bigcup$.

Typed Racket features polymorphic types. Polymorphic types in Typed Racket are very explicit:

```
(: twice (All (A) (-> A (Pairof A A))))
(define (twice a)
  (cons a a))
```

`cons` is the constructor for the `Pairof` type. You can read the type annotation as "for all types `A`, the type of `twice` is such that iff you pass it a value of some type `A` it will return a pair of type `A` × `A`".

6 Immutable State and Recursion

In purely functional languages, you cannot alter the state of the program. You cannot assign to a variable once it has been defined[2]. Consider this simple Java

[2] Actually, this is possible in Racket, but using side-effects makes everything much more complicated.

program:

```
public static bool isEven(int n) {
  while (1 < n)
    n = n - 2;
  return n == 0;
}
```

The function decrements `n` until it is less or equal to one. Then, it check whether `n` is equal to zero and returns the value of that check. While this is a very simple way to do it, because recursion in Java is costly, we cannot do it that way in Racket, because there, `n` can never be changed. Instead, we have to chose the recursive approach:

```
(: is-even? (-> Integer Boolean))
(define (is-even? n)
  (if (< 1 n)
      (is-even? (- n 2))
      (= n 0)))
```

Every time, `is-even?` is called with an `n` that is greater than 1, it calls itself with `(- n 2)`. The result is the same, but this way of thinking enables us to decompose a function into its *cases*!

7 Pattern Matching

Pattern matching is a very powerful tool of functional programming. We will use it time and time again to implement functions. Here is a variant of `is-even?` that uses pattern matching:

```
(: is-even? (-> Integer Boolean))
(define (is-even? n)
  (match n
    [0 #t]
    [1 #f]
    [_ (is-even? (- n 2))]))
```

The principle of the function stays the same: if the value is zero, then it is an even number. If it is one, it is an odd number. In any other case (this is what `_` means), the function calls itself recursively and decrements `n` by two. The wildcard `_` matches any value and it must always come last in the list of patterns, otherwise it will take over all following patterns, too.

By the way, you can use `_` anywhere you like if you do not care about the value of something.

8 Anonymous Functions

An anonymous function is a function without a name. You can create it and apply it immediately, as in this example:

```
> ((lambda (x) x) 2)
2
```

Here, we have defined a lambda expression and applied it to the value 2. We call this function the *identity function*. The line above is equal to $((\lambda\, x\, .\, x)\, 2)$.

9 The Maybe Type

The `Maybe` type is useful to model the absence of a value without having to explicitly check for null-pointers:

```
(struct None ())
(struct (A) Some ([a : A]))
(define-type (Maybe A) (U None (Some A)))
```

The type of `Maybe` is the union of two structs and it is polymorphic! `A` is just a placeholder, a generic name for a type. It is simlilar, but not equal, to this Java code:

```
public static abstract class Maybe<A> {}

public static class None<A> extends Maybe<A> {
  public None() {}
}

public static class Some<A> extends Maybe<A> {
  public final A a;
  public Some(A a) {
    this.a = a;
  }
}
```

Now, we can implement a function that performs safe division:

```
(: maybe-divide (-> Number Number (Maybe Number)))
(define (maybe-divide n d)
  (if (= d 0)
      (None)
      (Some (/ n d))))
```

We first check whether the divisor `d` is zero. If yes, then we cannot divide, so we return `None`. Otherwise, we return `Some (\ n d)`. This has many benefits. For the first, we do not need to catch any exceptions. Exceptions can go unnoticed when you write your code and then it later becomes a problem in production.

Secondly, the type of `maybe-divide` indicates that the computation can go wrong! This is useful information if you are writing an API and want to tell the user of the API that there is a chance for this function to fail. This might be anything from simple arithmetic to HTTP-requests.

We can check whether a `Maybe` has a value:

```
(: has-value? (-> (Maybe Any) Boolean))
(define (has-value? m)
  (match m
    [(None) #f]
    [(Some _) #t]))
```

We use pattern matching do *decompose* the value of `m` in order to see how it has been constructed and what is inside. Note that we do not care what value it contains, so we use `_`.

We can use a *higher-order function* – that is a function that takes another function as an argument – to only execute a function on a `Maybe` if it contains a value:

```
(: maybe-map (All (A B) (-> (-> A B)
                            (Maybe A)
                            (Maybe B))))
(define (maybe-map f m)
  (match m
    [(None) (None)]
    [(Some a) (Some (f a))]))
```

`maybe-map` is a polymorphic function, because it does not care what the *actual type* of `A` and `B` is.

`f` is of type `(-> A B)`. This is the function we want to apply if there is a value present in the maybe. It is important that the types match. Because `f` accepts a type `A` and returns a type `B`, the type of the second parameter to `maybe-map` must be of type `(Maybe A)` and the return-type must be of type `(Maybe B)`.

Now, we can use `maybe-map` together with a *lambda expression*:

```
> (maybe-map (lambda ([x : Number]) (* 2 x))
             (maybe-divide 42 23))
- : (U None (Some Number))
(Some 84/23)
```

10 Single-Linked Lists

Let us try to implement a more interesting data type. Lists are very useful, so let us implement the simplest list possible, a single-linked list. We need an empty list constructor, which we call `Nil` and a constructor for adding an element to a list, which we call `Cons`:

```
(struct Nil ())
(struct (A) Cons ([head : A] [tail : (LinkedList A)]))
(define-type (LinkedList A) (U Nil (Cons A)))
```

Let us instantiate such a list:

```
(Cons 'a
      (Cons 'b
            (Cons 'c
                  (Cons 'd
                        (Cons 'e (Nil))))))
```

This is how the list looks like, conceptually:



Every list instance contains another, smaller list instance. We call the value of a `Cons` cell `head` and the remaining list `tail`.

Let us see how we can compute the length of a list. We can use pattern-matching:

```
(: list-length (All (A) (-> (LinkedList A) Integer)))
(define (list-length as)
  (match as
    [(Nil) 0]
    [(Cons _ tail) (+ 1 (list-length tail))]))
```

There are two cases: either, the list is empty. In this case, the length is zero. Or the list has at least one element. Then, the length is one plus the length of the `tail`. Again, note that we do not care about the value of the `Cons` cell.

We can easily implement a function to push a new element to the list:

```
(: list-push (All (A) (-> A (LinkedList A)
                            (LinkedList A))))
(define (list-push a as)
  (Cons a as))
```

It takes constant time and space, because we re-use the old list. Indeed, many lists can share the same sub-list. For instance, all lists share the empty list `Nil`. This is legal, because lists are also immutable[3]!

Let us look at more functions on lists. For example, we would like to be able to concatenate two lists. We can again implement this conveniently by pattern-matching.

```
(: list-concat (All (A) (-> (LinkedList A)
                            (LinkedList A)
                            (LinkedList A))))
(define (list-concat lhs rhs)
```

[3]Sometimes people also call cons-lists a *persistent data structure*.

```
  (match lhs
    [(Nil) rhs]
    [(Cons head tail) (Cons head
                            (list-concat tail rhs))]))
```

Again, we perform case analysis: if the left hand side `lhs` is empty, then there is nothing to do and returning whatever is on the right hand side, `rhs`, is correct. If however `lhs` is not empty, we must add its `head` to the new list. How do we construct a new list? By calling `list-concat` on its `tail`.

Why is this correct? Because either, `tail` is `Nil`. In this case, `concat` will return `rhs`. So we have added `head` to `rhs`. This is, what we wanted. Or, `tail` is not empty. Then we add its current `head` to whatever is returned by `concat` on its `tail`. At some point, there is no `tail` left, because lists cannot be infinite (computers have finite memory), and we end up again at the base case.

Note, that this takes $O(n)$ time and space, where n is the length of `lhs`. We have to look at every element once but we also have to construct n new cells, because we cannot simply replace the empty list at the very end of `lhs` by `rhs`: lists are also immutable!

The `list-remove` function is a nice example to illustrate sub-list sharing. If we know that an element is inside the list, we can ask `list-remove` to remove it:

```
> (list-remove
     'c (Cons 'a
              (Cons 'b
                    (Cons 'c
                          (Cons 'd
                                (Cons 'e (Nil)))))))

(Cons 'a (Cons 'b (Cons 'd (Cons 'e #<Nil>))))
```

How does it do that? It simply traverses the list and builds new `Cons` cells until it encounters `'c`. Then it skips `'c` and returns the remaining list:



This is the code:

```
(: list-remove (All (A) (-> A (LinkedList A)
                            (LinkedList A))))
(define (list-remove a as)
  (match as
    [(Nil) (Nil)]
```

```
    [(Cons head tail)
       (if (eq? head a)
           tail
           (Cons head
           (list-remove a tail)))]))
```

11 Higher-Order Functions on Lists

Often times, we want to perform the same computation for all values that are stored in a container, such as a list. In functional programming, this can easily be done by using higher-order functions. We have already seen higher order functions for the `Maybe` type.

The function that allows us to apply the same function to all elements of a list is typically called `map`, because it maps some function to all values that the list contains. If we had no higher-order functions, we would need to write a new function that explicitly traverses the list, every time that we want to modify all list elements. `map` makes this unnecessary, because `map` rather focuses on the *shape of the data structure* than on whatever function will be executed.

This is our code for `map`:

```
(: list-map (All (A B) (-> (-> A B)
                           (LinkedList A)
                           (LinkedList B))))
(define (list-map f as)
  (match as
    [(Nil) (Nil)]
    [(Cons head tail) (Cons (f head)
                            (list-map f tail))]))
```

`map` does not care what the function `f` does, as long as it is of type `(-> A B)`. It is perfectly valid for `A` and `B` to be of the same type. Therefore, we could use a function `f` of type `(-> Integer Integer)`.

Here is the example from the lecture: a list of students and a function `better-grade` that increases the grade of a student by 10%:

```
(struct Student ([name : String]
                 [grade : Integer]))

(define students
        (Cons (Student "A" 89)
              (Cons (Student "B" 44)
                    (Cons (Student "C" 62) (Nil)))))

(: better-grade (-> Student Student))
(define (better-grade student)
  (Student (Student-name student)
           (+ 10 (Student-grade student))))
```

Recall that we can access any field of a struct by calling a function that is named after the pattern `<struct-name>-<field-name>`. We can now either write a new function that applies `better-grade` to every student in a list of students, or we use `map` to do so. Here are both variants:

```
(: everyone-better-grade (-> (LinkedList Student)
                             (LinkedList Student)))
(define (everyone-better-grade students)
  (match students
    [(Nil) (Nil)]
    [(Cons student tail)
       (Cons (better-grade student)
             (everyone-better-grade tail))]))
```

Note the similarity between `everyone-better-grade` and `list-map`.

```
> (everyone-better-grade students)
- : (U Nil (Cons Student))
(Cons (Student "A" 99)
      (Cons (Student "B" 54)
            (Cons (Student "C" 72) #<Nil>)))

> (list-map better-grade students)
- : (U Nil (Cons Student))
(Cons (Student "A" 99)
      (Cons (Student "B" 54)
            (Cons (Student "C" 72) #<Nil>)))
```

The result is the same, but we only have to write `map` once and we can use it on any type of list with any type of function (as long as the functions type is a specialization of the most general type `(-> A B)` or $A \to B$).

If we now want to compute the average of all grades for the course, we need to write a function that first extracts all grades from the students and then adds them together and then finally divides the sum of all grades by the number of students in the course. We could use the following code to get all grades:

```
> (list-map Student-grade students)
- : (U Nil (Cons Integer))
(Cons 89 (Cons 44 (Cons 62 #<Nil>)))
```

How do we sum up the resulting grades? We could write a simple sum function, but after we now have learned that we can use higher-order functions to implement re-usable solutions, maybe we can find a way to implement such a function to compute sums. Indeed, such functions exist and one of them is called `fold`. It takes a function of type `(-> B A B)`, a state parameter of type `B` and an input list of type `LinkedList A`:

```
(: list-fold (All (A B) (-> (-> B A B)
                            B
```

```
                              (LinkedList A) B)))
(define (list-fold f state as)
  (match as
    [(Nil) state]
    [(Cons head tail) (list-fold f
                                 (f state head)
                                 tail)]))
```

`state` is a parameter that carries the result from the last computation to the next recursive step. We have to initialize it in some way to make sure that we get an answer in case that the input list is empty. Let's try to fold the list of grades with `+` (which is of type `(-> Integer Integer Integer)` and therefore matches `fold`'s type requirement) and see what happens:

```
> (list-fold + 0 (list-map Student-grade students))
- : Integer
195
```

We start with zero as the initial state and add every student grade to state. Try, by hand, to take the `+` operator, the initial state and the list of grades and to place them in the definition of `fold` and see what happens at each step!

Our final function to compute the average of the course could look like this:

```
(: course-average (-> (LinkedList Student) Integer))
(define (course-average students)
  (/ (list-fold + 0 (list-map Student-grade students))
     (list-length students)))
```

This is a very concise way of writing this a function, thanks to the higher-order functions we have defined.

Finally, we want to only keep students in the list that have passed the course. Again, we can either write a new function that decomposes the list step by step or we write a higher-order function that removes all elements of a list for which a certain predicate `p` holds. Predicates are always of the type `(-> A Boolean)`. We call this function `filter` and we can define it as follows:

```
(: list-filter (All (A) (-> (-> A Boolean)
                            (LinkedList A)
                            (LinkedList A))))
(define (list-filter p as)
  (match as
    [(Nil) (Nil)]
    [(Cons head tail)
       (if (p head)
           (list-filter p tail)
           (Cons head
                 (list-filter p tail)))]))
```

The `filter` functions follows the same principle as `remove`, but it does not stop after the first element it encounters. It keeps going and calls `p` for every

element of the list until it reaches the end of the list. We can call it using a lambda expression that returns true if a grade is below 50%:

```
> (list-filter (lambda ([g : Integer]) (< g 50))
               (list-map Student-grade students))
- : (U Nil (Cons Integer))
(Cons 89 (Cons 62 #<Nil>))
```